# Differential Equations

James Rohal

First we load these packages.

```
> with(DETools):
  with(plots):
```

We will be studying the differential equation $y'(t) = \frac{1}{2}\left(y(t)^2 - 1\right)$ today where $t$ is the independent variable and $y$ is the dependent variable. Sometimes this is written as $y' = \frac{1}{2}\left(y^2 - 1\right)$ when it is understood that $y$ is the dependent variable in the differential equation. When represented this way, it doesn't necessarily matter what we choose our independent variable to be.

To represent a
differential equation in Maple, we need to explicitly tell it what the independent and dependent variables are. We will choose our independent variable to be $t$.

```
> ode := diff(y(t), t) = (1/2)*(y(t)^2 - 1);
```

$$ode := \frac{d}{dt}\, y(t) = \frac{1}{2}\, y(t)^2 - \frac{1}{2} \tag{1}$$

First, we show $y(t) = \frac{1 + ce^t}{1 - ce^t}$ is a solution to this differential equation. (Here $c$ is a constant.) This means we need to plug $y(t)$ into our differential equation. We work on the left hand side first. Then the right hand side.

```
> lhs_ode:= eval(lhs(ode), y(t) = (1+c*exp(t))/(1-c*exp(t)));

  lhs_ode := simplify(lhs_ode);
```

$$lhs\_ode := \frac{c\, e^t}{1 - c\, e^t} + \frac{\left(1 + c\, e^t\right) c\, e^t}{\left(1 - c\, e^t\right)^2}$$

$$lhs\_ode := \frac{2\, c\, e^t}{\left(-1 + c\, e^t\right)^2} \tag{2}$$

```
> rhs_ode:= eval(rhs(ode), y(t) = (1+c*exp(t))/(1-c*exp(t)));

  rhs_ode := simplify(rhs_ode);
```

$$rhs\_ode := \frac{1}{2}\, \frac{\left(1 + c\, e^t\right)^2}{\left(1 - c\, e^t\right)^2} - \frac{1}{2}$$

$$rhs\_ode := \frac{2\, c\, e^t}{\left(-1 + c\, e^t\right)^2} \tag{3}$$

As we see, **lhs_ode** and **rhs_ode** agree. Hence $y(t) = \frac{1 + ce^t}{1 - ce^t}$ is a solution to this differential equation for any $c$. Let's get Maple to solve this differential equation for us.

```
> ode_soln := dsolve(ode, y(t));
  simplify(eval(-(1-exp(-2*x))/(1+exp(-2*x)) ,x=(1/2)*t+(1/2)*_C1))
```

```
;
```

$$ode\_soln := y(t) = -\tanh\left(\frac{1}{2}\,t + \frac{1}{2}\,\_C1\right)$$

$$\frac{-1 + e^{-t - \_C1}}{1 + e^{-t - \_C1}} \quad \textbf{(4)}$$

The $\_C1$ is Maple's way of representing a constant.

Let's study this differential equation by looking at the direction field. Bigger arrows mean the "flow is stronger" there.

```
> fieldplot([1,(1/2)*(y^2 - 1)], t=-5..5, y=-5..5, arrows=thick,
  grid=[30,30], fieldstrength=fixed(0.5));
```




It looks like there is some kind of equilibrium solution between $y = -2$ and $y = 2$. Turns out there are two equilibrium solutions at $y(t) = -1$ and $y(t) = 1$. We can verify this by finding the equilbirum solutions, which occur when $\frac{dy}{dt} = 0$.

```
> solve(rhs(ode) = 0, y(t));
```

$$1,\ -1 \quad \textbf{(5)}$$

We can then plot them in red on our vector field.

```
> equilirium_1_plot := plot(-1, t=-5..5, color=red, thickness=2):
```

```
equilirium_2_plot := plot(1, t=-5..5, color=red, thickness=2):

field             := fieldplot([1,(1/2)*(y^2 - 1)], t=-5..5, y=
-5..5, arrows=thick, grid=[30,30], fieldstrength=fixed(0.5)):

# show all the plots together
display(field, equilirium_1_plot, equilirium_2_plot);
```




By studying the direction of the arrows we can see that both equilbriums are stable.

Let's plot a few more solutions by choosing different initial conditions.

```
> soln_1      := dsolve({ode, y(0)=1.009}, y(t), type=numeric,
range=-5..5):
soln_1_plot := odeplot(soln_1, color=yellow, thickness=2):

soln_2      := dsolve({ode, y(0)=0.8}, y(t), type=numeric, range=
-5..5):
soln_2_plot := odeplot(soln_2, color=green, thickness=2):

soln_3      := dsolve({ode, y(0)=0}, y(t), type=numeric, range=
-5..5):
soln_3_plot := odeplot(soln_3, color=blue, thickness=2):

soln_4      := dsolve({ode, y(0)=-0.8}, y(t), type=numeric,
```

```
range=-5..5):
soln_4_plot := odeplot(soln_4, color=violet, thickness=2):

soln_5      := dsolve({ode, y(0)=-1.009}, y(t), type=numeric,
range=-5..5):
soln_5_plot := odeplot(soln_5, color=black, thickness=2):

display(field, soln_1_plot, soln_2_plot, soln_3_plot,
soln_4_plot, soln_5_plot, equilirium_1_plot, equilirium_2_plot);
```




Let's study one of the solutions in a bit more detail and show that it approaches the equlibrium solutions. Consider the differential equation with initial condition $y(0) = 0.8$. This is the green curve above. We first find the solution.

```
> soln_with_initial := dsolve({ode, y(0)=0.8}, y(t));
```

$$soln\_with\_initial := y(t) = -\tanh\left(\frac{1}{2}\, t - \operatorname{arctanh}\left(\frac{4}{5}\right)\right) \tag{6}$$

Now let's compute the limit as $t$ tends to $\pm\infty$.

```
> limit(soln_with_initial, t=infinity);

limit(soln_with_initial, t=-infinity);
```

$$\lim_{t \to \infty} y(t) = -1$$

**(7)**

$$\lim_{t \to -\infty} y(t) = 1 \quad (7)$$

# Homework

You will be studying the logistic equation $\frac{dP}{dt} = kP\left(1 - \frac{P}{M}\right)$.

## (a) Assign the ODE to `ode`

## (b) Verify that $P(t) = \frac{M}{1 + Ae^{-kt}}$, $A = \frac{M - P0}{P0}$ is a solution to the logistic equation by using `eval` and `simplify`

## (c) Find the equilibrium solutions using `solve`

## (d) Let $M = 1000$, $k = 0.08$. Plot the equilibrium solutions and the direction field on `t=0..100` and `P=0..1100` on the same plot. This requires the use of `display`, `plot`, and `fieldplot`.

## (e) Plot the solution using the initial condition `P(0) = 80` with `range=0..100` by using `dsolve` with `type=numeric` and `odeplot`. Plot the solution on the same plot as the direction field and the equilbrium solutions.

## (f) Let `P(t)` be the solution with initial condition `P(0) = 80`. Find $\lim_{t \to +\infty} P(t)$ and $\lim_{t \to -\infty} P(t)$ using `limit`.